// File: /contracts/ThePirateBay.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.4;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC721/IERC721Receiver.sol";
import "@openzeppelin/contracts/token/ERC721/IERC721.sol";
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/security/Pausable.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";

interface ICryptoPirates {
    function types(uint256) external view returns (uint256);
    function TYPE_COMMON() external view returns (uint256);
    function TYPE_RARE() external view returns (uint256);
    function TYPE_EPIC() external view returns (uint256);
    function TYPE_LEGENDARY() external view returns (uint256);
}

contract ThePirateBay is Ownable, IERC721Receiver, Pausable, ReentrancyGuard {
    // whether or not rescue is enabled
    bool public rescueEnabled = false;
    // address of the NFT contract
    IERC721 public nftAddress;
    // address of the ERC20 token
    IERC20 public tokenAddress;

    uint256 public constant amountPerNFT = 10_000_000 ether; // amount of tokens per NFT
    uint256 public constant totalLockTime = 10 days; // how long the lock is
    uint256 public constant totalReward = 2_000_000_000 ether; // total reward for the pool (2 billion tokens)

    uint256 public constant COMMON_MULTIPLIER = 100; // 1
    uint256 public constant RARE_MULTIPLIER = 125; // 1.25
    uint256 public constant EPIC_MULTIPLIER = 200; // 2
    uint256 public constant LEGENDARY_MULTIPLIER = 500; // 5
    
    bool public depositStarted = true; // whether or not depositing has started
    uint256 public lockStartTime = 1686650400; // Tuesday, June 13, 2023 10:00:00 AM

    uint256 public totalStaked; // total amount of tokens staked
    uint256 public totalNFTsStaked; // total amount of NFTs staked
    uint256 public totalMultipliers; // total amount of multipliers

    struct Stake {
        uint256 amount; // amount of tokens staked
        uint256[] nftIds; // array of NFTs staked
        bool claimed; // whether the user has claimed their reward
    }

    mapping(uint256 => bool) private nftIdsStaked; // whether or not an NFT has been staked

    mapping(address => Stake) public stakes; // stake of an account

    event Staked(address indexed user, uint256 amount, uint256[] nftIds);
    event Claimed(address indexed user, uint256 amount);

    constructor(address _nftAddress, address _tokenAddress) {
        nftAddress = IERC721(_nftAddress);
        tokenAddress = IERC20(_tokenAddress);
    }

    modifier whenDepositOn() {
        require(depositStarted, "Cannot stake before deposit started");
        require(lockStartTime > block.timestamp, "Cannot stake after lock started");
        _;
    }

    modifier whenLockOver() {
        require(lockStartTime + totalLockTime < block.timestamp, "Cannot claim during lock time");
        _;
    }

    modifier whenRescueEnabled() {
        require(rescueEnabled, "Rescue is not enabled");
        _;
    }

    /// @dev allows user to stake tokens and NFTs
    /// @param _nftIds ids of the NFTs to stake
    function stake(uint256[] calldata _nftIds) external whenNotPaused whenDepositOn nonReentrant {
        require(_nftIds.length > 0, "Cannot stake 0 NFTs");
        
        uint256 totalAmount = amountPerNFT * _nftIds.length;

        Stake storage _stake = stakes[_msgSender()];

        // transfer tokens to contract
        tokenAddress.transferFrom(_msgSender(), address(this), totalAmount);
        // transfer NFTs to contract
        for (uint256 i = 0; i < _nftIds.length; i++) {
            require(!nftIdsStaked[_nftIds[i]], "Cannot stake the same NFT twice");
            nftAddress.transferFrom(_msgSender(), address(this), _nftIds[i]);
            _stake.nftIds.push(_nftIds[i]);
            nftIdsStaked[_nftIds[i]] = true;
            totalMultipliers += nftMultiplier(_nftIds[i]);
        }

        // update stake
        _stake.amount += totalAmount;

        // update totals
        totalStaked += totalAmount;
        totalNFTsStaked += _nftIds.length;

        emit Staked(_msgSender(), totalAmount, _nftIds);
    }

    /// @dev allows user to claim tokens and NFTs
    function claim() external whenNotPaused whenLockOver nonReentrant {
        Stake memory _stake = stakes[_msgSender()];
        require(_stake.amount > 0, "Cannot claim before staking");
        require(!_stake.claimed, "Cannot claim twice");

        _stake.claimed = true;

        // transfer NFTs to user
        for (uint256 i = 0; i < _stake.nftIds.length; i++) {
            nftAddress.transferFrom(address(this), _msgSender(), _stake.nftIds[i]);
        }

        uint256 totalAmount = _stake.amount + earned(_msgSender());

        // transfer tokens to user
        require(IERC20(tokenAddress).transfer(_msgSender(), totalAmount), "Cannot transfer tokens to user");

        emit Claimed(_msgSender(), totalAmount);
    }

    /// @dev returns the amount of tokens earned
    /// @param account address of the account
    function earned(address account) public view returns (uint256 _earned) {
        uint256 multiplier = totalMultiplier(account);
        return multiplier * totalReward / totalMultipliers;
    }

    /// @dev returns the multiplier for an NFT
    /// @param nftId id of the NFT
    function nftMultiplier(uint256 nftId) public view returns (uint256 multiplier) {
        ICryptoPirates cryptoPirates = ICryptoPirates(address(nftAddress));
        uint256 nftType = cryptoPirates.types(nftId);
        if (nftType == cryptoPirates.TYPE_COMMON()) {
            return COMMON_MULTIPLIER;
        } else if (nftType == cryptoPirates.TYPE_RARE()) {
            return RARE_MULTIPLIER;
        } else if (nftType == cryptoPirates.TYPE_EPIC()) {
            return EPIC_MULTIPLIER;
        } else if (nftType == cryptoPirates.TYPE_LEGENDARY()) {
            return LEGENDARY_MULTIPLIER;
        }
        return COMMON_MULTIPLIER;
    }

    /// @dev returns the total multiplier for an account
    /// @param account address of the account
    function totalMultiplier(address account) public view returns (uint256 multiplier) {
        Stake memory _stake = stakes[account];
        for (uint256 i = 0; i < _stake.nftIds.length; i++) {
            multiplier += nftMultiplier(_stake.nftIds[i]);
        }
        return multiplier;
    }

    /// @dev returns the NFTs staked by an account
    /// @param account address of the account
    function userNFTs(address account) public view returns (uint256[] memory nftIds) {
        Stake memory _stake = stakes[account];
        return _stake.nftIds;
    }

    /// @dev returns the stats of an account
    /// @param account address of the account
    function userStats(address account) public view 
        returns (uint256 nfts, uint256 staked, uint256 multiplier, uint256 share, uint256 reward, bool claimed)
    {
        Stake memory _stake = stakes[account];
        nfts = _stake.nftIds.length;
        staked = _stake.amount;
        multiplier = totalMultiplier(account);
        share = multiplier * 10000 / totalMultipliers;
        uint256 stakeTime = 0;
        if (lockStartTime > block.timestamp) stakeTime = 0;
        else if (lockStartTime + totalLockTime < block.timestamp) stakeTime = totalLockTime;
        else stakeTime = block.timestamp - lockStartTime;
        reward = earned(account) * stakeTime / totalLockTime;
        claimed = _stake.claimed;
    }

    /// @dev allows owner to set the lock start time
    /// @param _lockStartTime timestamp of when locking starts
    function setLockStartTime(uint256 _lockStartTime) external whenDepositOn onlyOwner {
        require(_lockStartTime > block.timestamp, "Cannot set time before current time");
        lockStartTime = _lockStartTime;
    }

    /// @dev allows owner to set wether or not depositing has started
    /// @param _depositStarted boolean
    function setDepositStarted(bool _depositStarted) external onlyOwner {
        depositStarted = _depositStarted;
    }

    /// @dev allows owner to enable "rescue mode"
    /// @param _enabled boolean
    function setRescueEnabled(bool _enabled) external onlyOwner {
        rescueEnabled = _enabled;
    }

    /// @dev enables owner to pause / unpause minting
    /// @param _paused boolean
    function setPaused(bool _paused) external onlyOwner {
        if (_paused) _pause();
        else _unpause();
    }

    /// @dev allows user to rescue tokens and NFTs
    /// @param account address of the account
    /// @param index index of the first NFT to rescue
    /// @param amount amount of NFTs to rescue
    function emergencyRescue(address account, uint256 index, uint256 amount) external whenNotPaused whenRescueEnabled nonReentrant {
        Stake memory _stake = stakes[account];

        // transfer NFTs to user
        for (uint256 i = index; i < (index + amount); i++) {
            nftAddress.transferFrom(address(this), account, _stake.nftIds[i]);
        }
    }

    /// @dev allows user to rescue tokens
    /// @param account address of the account
    function emergencyRescueTokens(address account) external whenNotPaused whenRescueEnabled nonReentrant {
        Stake memory _stake = stakes[account];
        require(!_stake.claimed, "Cannot rescue tokens after claiming");
        _stake.claimed = true;
        // transfer tokens to user
        require(IERC20(tokenAddress).transfer(account, _stake.amount), "Cannot transfer tokens to user");
    }


    /// @dev allows owner to rescue tokens
    /// @param amount amount of tokens to rescue
    function emergencyRewardRescue(uint256 amount) external onlyOwner whenRescueEnabled {
        require(IERC20(tokenAddress).transfer(_msgSender(), amount), "Cannot transfer tokens to user");
    }

    function onERC721Received(
        address,
        address from,
        uint256,
        bytes calldata
    ) external pure override returns (bytes4) {
      require(from == address(0x0), "Cannot send tokens directly to the bay");
      return IERC721Receiver.onERC721Received.selector;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721Receiver.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Note that the caller is responsible to confirm that the recipient is capable of receiving ERC721
     * or else they may be permanently lost. Usage of {safeTransferFrom} prevents loss, though the caller must
     * understand this adds an external call which potentially creates a reentrancy vulnerability.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/security/Pausable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (security/Pausable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which allows children to implement an emergency stop
 * mechanism that can be triggered by an authorized account.
 *
 * This module is used through inheritance. It will make available the
 * modifiers `whenNotPaused` and `whenPaused`, which can be applied to
 * the functions of your contract. Note that they will not be pausable by
 * simply including this module, only once the modifiers are put in place.
 */
abstract contract Pausable is Context {
    /**
     * @dev Emitted when the pause is triggered by `account`.
     */
    event Paused(address account);

    /**
     * @dev Emitted when the pause is lifted by `account`.
     */
    event Unpaused(address account);

    bool private _paused;

    /**
     * @dev Initializes the contract in unpaused state.
     */
    constructor() {
        _paused = false;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is not paused.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    modifier whenNotPaused() {
        _requireNotPaused();
        _;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is paused.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    modifier whenPaused() {
        _requirePaused();
        _;
    }

    /**
     * @dev Returns true if the contract is paused, and false otherwise.
     */
    function paused() public view virtual returns (bool) {
        return _paused;
    }

    /**
     * @dev Throws if the contract is paused.
     */
    function _requireNotPaused() internal view virtual {
        require(!paused(), "Pausable: paused");
    }

    /**
     * @dev Throws if the contract is not paused.
     */
    function _requirePaused() internal view virtual {
        require(paused(), "Pausable: not paused");
    }

    /**
     * @dev Triggers stopped state.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    function _pause() internal virtual whenNotPaused {
        _paused = true;
        emit Paused(_msgSender());
    }

    /**
     * @dev Returns to normal state.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    function _unpause() internal virtual whenPaused {
        _paused = false;
        emit Unpaused(_msgSender());
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}